
> > with(GraphTheory);

with(SpecialGraphs);

W5 := WheelGraph(5);

DrawGraph(W5);

IsVertexColorable(W5,3);
false

P := PetersenGraph();

DrawGraph(P);

The Petersen graph is 3 colorable.
IsVertexColorable(P,3,'C');

true

C;

HighlightVertex(P,C[1],red);

HighlightVertex(P,C[2],green);

DrawGraph(P);

Vertices(W5);

seq( x[u]^k-1, u=Vertices(W5) );

Edges(W5);

seq( x[e[1]]^k-x[e[2]]^k, e=Edges(W5) );

Sys := proc(G::Graph,x::name,k::nonnegint)
local V,E,v,e,S;
   V,E := Vertices(G), Edges(G);
   S := {seq( x[v]^k-1, v=V )} union
        {seq( normal( (x[e[1]]^k-x[e[2]]^k) / (x[e[1]]-x[e[2]]) 
), e=E )}
end:   

S := Sys( WheelGraph(3), x, 3 );

S := subs( x[0]=1, S );

Groebner[Basis]( S, tdeg(x[0],x[1],x[2],x[3]) ); 

S := Sys( WheelGraph(4), x, 3 );

S := subs( x[0]=1, S );

Groebner[Basis](S, tdeg(x[1],x[2],x[3],x[4]) );

Monomials := proc(X::set(name),d::nonnegint) option remember; 
local x,i,m;
   # return all monomials in X of degree 0,1,...,d .
   if X={} then return 1 fi;
   x := X[1];
   seq( seq( x^i*m, m=Monomials( X[2..-1], d-i ) ), i=0..d );
end:

Monomials( {x,y,z}, 3 );

HNSS := proc( S::set(polynom), d::nonnegint, c::name, 
ansatz::name )
local M,i,one,X;
    X := indets(S,name); #print(X);
    M := [Monomials(X,d)]; #print(M);
    one := add( add( c[i,j]*M[j], j=1..nops(M) )*S[i], i=1..nops
(S) );
    if nargs=4 then ansatz := one fi;
    { coeffs( expand(one)-1, X ) };
end:

Consider W(5)
S := HNSS( Sys(WheelGraph(5),x,3), 1, c, 'ANS' ); 

ANS;

Try to find a degree 1 certificate for W(5)
nops(S);

115

nops( indets(S) );
112

{solve(S)};

S := HNSS( Sys(WheelGraph(5),x,3), 2, c ): 

A degree 2 certificate
nops(S);

336

nops( indets(S) );
448

{solve(S)};

S := HNSS( Sys(WheelGraph(5),x,3), 3, c ):

A degree 3 certificate
nops(S);

783

nops(indets(S));
1344

sol := {solve(S)}: # should output {} but getting bad answer in 

Maple 17

S := HNSS( Sys(WheelGraph(5),x,3), 4, c, 'ZZ' ):

A degree 4 certificate
nops(S);

1590

nops(indets(S));
3360

sol := solve(S);
[Length of output exceeds limit of 1000000]

Certificate := proc(ansatz, sol, c) local p;
    p := map( proc(e) if lhs(e)=rhs(e) then lhs(e)=1 fi end, sol 
);
    subs( p, subs( sol, ansatz ) );
end:

Certificate( ZZ, sol, c ):

expand(%);
1

S := HNSS( Sys(WheelGraph(5),x,3), 1, c, 'ZZ' ): 

But a degree 1 certificate exists modulo 2 !!
sol := msolve( S, 2 );

Zvals := map( proc(e) if type(rhs(e),name) then rhs(e) fi end, 
sol );

one := subs( {seq( z=1, z=Zvals)}, subs( sol, ZZ ) ) mod 2;

Expand(%) mod 2;
1


